
/* 
2. ČÁST
V této části zhotovíme bodovou konstrukci řezu.
Jednotlivou bodovou konstrukcí je průnik kužele
s rovinou rovnoběžnou s půdorysnou - kružnice - 
a průnik této kružnice s rovinou řezu (průnik kružnice
s průsečnicí roviny řezu a roviny, ve které leží 
tato kružnice.
Výsledkem každé bodové konstrukce jsou dva body.
*/

#vloz "zakladni.dk"

// průnik s rovinou ve výšce z
rez(z) {
    X = S + bod(0,0,z); 
    alfa = rovnobezna(Pi, X);
    Y = p(alfa.2, (b:0).2); // průnik s bokem
    d = v(X.2, Y.2); // poloměr nové kružnice
    l = kruz(alfa, X, d); // řez kužele rovinou alfa
    
    r = p(ro, alfa); // průsečnice roviny řezu a alfa
    R = p(l, r); // výsledek bodové konstrukce
    
    vrat R;    
}

hlavni () {
    zobrazeni("mongeovo");
    papir(29.7, 21, 1); //A4
    okraj(0.75,0.75,0.75,0.75,0.25,1,0);
    posunutipss(0,0);    
    nadpis("", "", datum(), "", "ØEZ KUŽELE ROVINOU");

    // předdefinované styly
    normal = styl();    

    sklopeny = styl();
    videt(sklopeny, vse, 2, ?, ?, 1, 3);
    barva(sklopeny, "seda");

    silny = styl();
    videt(silny, vse, 1, 0, 3, 2, 0);

    osa = styl();
    videt(osa, vse, 1, 0, 3, 1, 2);

    schovany = styl();
    videt(schovany, vse, 1, 0, 2, 1, 3);
    
    nevidet = styl();
    videt(nevidet, vse, 0);
    
    //kuzel
    S = bod(2,6,0); // střed podstavy
    r = 4; // poloměr podstavy
    v = 9; // výška kužele

    k = kruz(Pi, S, r); // podstava
    o = kolmice(Pi, S); // osa
    V = S + bod(0,0,v); // vrchol 



    // boční čáry
    K = bod(x, S, r);      
    b = usecka(K, V);

    styl(o, osa);
    styl(m(k, b), silny);

    videt(b, 1, 0);
    videt(K, 1, 0);
    popis(K, 2, 0);

    // rovina
    ro = rovina(-9,9,8);
    
    i = 0; // začínáme zdola
    krok = 0.1; // přesnost

    dokud (i < v-krok) { // vrchol vynecháme
        R = rez(i);
        zvetsi(i, krok);
    }
    zrus(i);
    popis(R, vse, 0);
}


